



Cuadernillo de examen

ASIGNATURA	PROGRAMAS DE APLICACIÓN II PROGRAMAS DE APLICACIÓN III	CÓDIGO	305 305
CONVOCATORIA	Junio 1999 (2º Parcial)	PLAN DE ESTUDIOS	1994/1996
ESPECIALIDAD	Común	CURSO	3º
TURNO	Tarde	CENTRO	Facultad
CARÁCTER	Anual	CURSO ACADÉMICO	1998/99
DURACIÓN APROXIMADA	2 horas		

APELLIDOS: NOMBRE:

GRUPO: NÚMERO DE EXPEDIENTE:

Parte teórica

- Tipos de componentes ActiveX que se pueden desarrollar en Visual Basic 5.0.
- Características de los documentos ActiveX (creación, implementación, depuración y elementos que aparecen tras la compilación)
- El objeto Recordset ¿En qué consiste? ¿Qué tipos de recordset se contemplan y en que se diferencian en el modelo del DAO?
- VBScript. Explique las diferencias más significativas entre Visual Basic y VBScript ¿Cómo se integra un guión VBScript en una página HTML?

Puntuación: 1 punto cada pregunta

Parte práctica

- Una distribuidora editorial tiene los datos almacenados en una base de datos Access (libros.mdb) y desea gestionar las editoriales con las que trabaja. Para ellos, la base de datos dispone entre otras de las siguientes tablas:

Tabla: Editores		
Nombre Campo	Formato	Observaciones
IDEditor	Entero largo	Clave primaria. Identifica unívocamente al editor.
Nombre	Texto, 50 posiciones	Nombre de la editorial

Tabla: Titulos		
Nombre Campo	Formato	Observaciones
ISBN	Texto, 20 posiciones	Clave primaria. Identifica unívocamente al libro
Título	Texto, 50 posiciones	Título de la obra
IDEditor	Entero largo	Identificador del editor que publica la obra
AñoPublicacion	Entero	Año de publicación de la obra

Tabla: Autores		
Nombre Campo	Formato	Observaciones
IDAutor	Entero largo	Clave primaria. Identifica unívocamente a un autor
Autor	Texto, 50 posiciones	Nombre del autor
AñoPublicacion	Entero	Año de nacimiento del autor

Tabla: TituloAutor (Tabla de relación entre autores y obras)		
Nombre Campo	Formato	Observaciones
ISBN	Texto, 20 posiciones	ISBN de la obra. Forma el índice idxISBN (con duplicados)
IDAutor	Entero largo	Identificador del autor. Forma el índice idxIDAutor (con duplicados)



Cuadernillo de examen


ASIGNATURA	PROGRAMAS DE APLICACIÓN II PROGRAMAS DE APLICACIÓN III	CÓDIGO	305 305
CONVOCATORIA	Junio 1999 (2º Parcial)	PLAN DE ESTUDIOS	1994/1996
ESPECIALIDAD	Común	CURSO	3º
TURNO	Tarde	CENTRO	Facultad
CARÁCTER	Anual	CURSO ACADÉMICO	1998/99
DURACIÓN APROXIMADA	2 horas		

APELLIDOS: NOMBRE:

GRUPO: NÚMERO DE EXPEDIENTE:

Nota: Las tablas están relacionadas. Las relaciones realizadas hacen la actualización y eliminación de registros en cascada.

La distribuidora desea implementar un formulario como parte de una aplicación que realice el mantenimiento de las editoriales. El formulario tendrá el siguiente aspecto:



Notas importantes: Sólo existen los controles que aparecen en el formulario.

Todos los accesos a datos se deben hacer mediante objetos del DAO.

Se deben codificar los procedimientos y declaraciones necesarios para:

- A. Al pulsar el botón Buscar, se buscará la editorial que el usuario haya introducido en el campo txtIDEditor. Si éste no existiera, aparecerá un mensaje advirtiéndolo. Si existe, se recuperarán en un recordset los títulos de dicha editorial, apareciendo en los campos del *frame* Títulos. Si el editor no tuviera títulos aparecería también un mensaje.
Puntuación: 1,25 puntos.
- B. Gestionar el movimiento por los títulos del recordset mediante los botones Primero, Ultimo, Siguiete y Anterior.
Puntuación: 1 punto.
- C. Al pulsar el botón Borrar, se borrará al editor, todas sus obras y los registros correspondientes de la tabla TituloAutor.
Puntuación: 1,25 puntos.

Solución

Option Explicit



Cuadernillo de examen

ASIGNATURA	PROGRAMAS DE APLICACIÓN II PROGRAMAS DE APLICACIÓN III	CÓDIGO	305 305
CONVOCATORIA	Junio 1999 (2º Parcial)	PLAN DE ESTUDIOS	1994/1996
ESPECIALIDAD	Común	CURSO	3º
TURNO	Tarde	CENTRO	Facultad
CARÁCTER	Anual	CURSO ACADÉMICO	1998/99
DURACIÓN APROXIMADA	2 horas		

APELLIDOS: NOMBRE:

GRUPO: NÚMERO DE EXPEDIENTE:

```
Dim db As Database
Dim rs As Recordset
Dim rsEditores As Recordset
```

```
Private Sub cmdAnterior_Click()
    rs.MovePrevious
    If rs.BOF Then
        MsgBox "No hay más títulos", vbInformation, "Obras por editor"
        rs.MoveFirst
    Else
        CargarCampos
    End If
End Sub
```

```
Private Sub cmdBorrar_Click()
    Dim SQL As String

    SQL = "DELETE FROM Editores WHERE IDEditor = " & txtIDEditor
    db.Execute SQL
End Sub
```

```
Private Sub cmdBuscar_Click()
    Dim Qry As String

    'Buscar el nombre del editor
    Set rsEditores = db.OpenRecordset("Editores")
    rsEditores.Index = "idxIDEditor"
    rsEditores.Seek "=", txtIDEditor
    If rsEditores.NoMatch Then
        MsgBox "El editor no existe", vbInformation, "Obras por editor"
        txtIDEditor = Empty
        txtNombre = Empty
        Exit Sub
    Else
        txtNombre = rsEditores("Nombre")
    End If
```

```
'Carga en rs los títulos del editor
Qry = "SELECT Titulos.Titulo, Autores.Autor, Titulos.AñoPublicacion "
Qry = Qry & "FROM Titulos,Autores,TituloAutor,Editores "
Qry = Qry & "WHERE Titulos.ISBN = TituloAutor.ISBN "
Qry = Qry & " AND Autores.IDAutor = TituloAutor.IDAutor "
Qry = Qry & " AND Titulos.IDEditor = " & txtIDEditor

Set rs = db.OpenRecordset(Qry)
```



Cuadernillo de examen

ASIGNATURA	PROGRAMAS DE APLICACIÓN II PROGRAMAS DE APLICACIÓN III	CÓDIGO	305 305
CONVOCATORIA	Junio 1999 (2º Parcial)	PLAN DE ESTUDIOS	1994/1996
ESPECIALIDAD	Común	CURSO	3º
TURNO	Tarde	CENTRO	Facultad
CARÁCTER	Anual	CURSO ACADÉMICO	1998/99
DURACIÓN APROXIMADA	2 horas		

APELLIDOS: NOMBRE:

GRUPO: NÚMERO DE EXPEDIENTE:

```
    If rs.RecordCount = 0 Then
        MsgBox "El editor no tiene títulos", vbInformation, "Obras por editor"
    Else
        CargarCampos
    End If

End Sub

Private Sub cmdPrimero_Click()
    rs.MoveFirst
    CargarCampos
End Sub

Private Sub cmdSiguiete_Click()
    rs.MoveNext
    If rs.EOF Then
        MsgBox "No hay más títulos", vbInformation, "Obras por editor"
        rs.MoveLast
    Else
        CargarCampos
    End If
End Sub

Private Sub cmdUltimo_Click()
    rs.MoveLast
    CargarCampos
End Sub

Private Sub Form_Load()
    Set db = DBEngine.Workspaces(0).OpenDatabase("\\luis\bd\libros.mdb")
End Sub

Sub CargarCampos()
    txtTitulo = rs("Titulo")
    txtAutor = rs("Autor")
    txtAño = rs("AñoPublicacion")
End Sub
```

2. Implemente un control ActiveX (SoundButton) que permita reproducir un sonido cuando se pulse sobre él. Deberá escribir el código de todo el formulario (UserControl) que permite desarrollar este control. Este estará compuesto únicamente por un control CommandButton. El alumno deberá implementar la propiedad Caption (la etiqueta del botón) y la propiedad FileName con la vía de acceso y el nombre del archivo WAV del sonido que se va a reproducir. Así mismo debe hacer público el evento Click del botón y todas las declaraciones que considere oportunas.

Nota: La función API para reproducir un archivo WAV se encuentra en la biblioteca winmm.dll y es:



Cuadernillo de examen

ASIGNATURA	PROGRAMAS DE APLICACIÓN II PROGRAMAS DE APLICACIÓN III	CÓDIGO	305 305
CONVOCATORIA	Junio 1999 (2º Parcial)	PLAN DE ESTUDIOS	1994/1996
ESPECIALIDAD	Común	CURSO	3º
TURNO	Tarde	CENTRO	Facultad
CARÁCTER	Anual	CURSO ACADÉMICO	1998/99
DURACIÓN APROXIMADA	2 horas		

APELLIDOS: NOMBRE:

GRUPO: NÚMERO DE EXPEDIENTE:

```
sndPlaySound (ByVal lpszSoundName As String, ByVal uFlags As Long) As Long
```

donde:

- lpszSoundName es una cadena que contiene el nombre del archivo WAV a reproducir
- uFlag indica el modo de reproducción (1 para modo asíncrono que es el que debemos utilizar)

Puntuación: 2,5 puntos

Solución

```
Option Explicit

Const m_def_FileName = "c:\windows\media\ding.wav"

Dim m_FileName As String

Event Click()

Private Declare Function sndPlaySound Lib "winmm.dll" _
    Alias "sndPlaySoundA" _
    (ByVal lpszSoundName As String, ByVal uFlags As Long) As Long

Private Sub Command1_Click()
    Dim Retorno As Long

    Retorno = sndPlaySound(fileName, 1)
End Sub

Private Sub UserControl_InitProperties()
    Command1.Caption = Extender.Name
    m_FileName = m_def_FileName
End Sub

Public Property Get FileName() As String
    FileName = m_FileName
End Property

Public Property Let FileName(ByVal New_FileName As String)
    m_FileName = New_FileName
    PropertyChanged "FileName"
End Property

' Cargar valores de propiedades desde el almacenamiento
Private Sub UserControl_ReadProperties(PropBag As PropertyBag)
    m_FileName = PropBag.ReadProperty("FileName", m_def_FileName)
    Command1.Caption = PropBag.ReadProperty("Caption", "Command1")
End Sub
```



Cuadernillo de examen

ASIGNATURA	PROGRAMAS DE APLICACIÓN II PROGRAMAS DE APLICACIÓN III	CÓDIGO	305 305
CONVOCATORIA	Junio 1999 (2º Parcial)	PLAN DE ESTUDIOS	1994/1996
ESPECIALIDAD	Común	CURSO	3º
TURNO	Tarde	CENTRO	Facultad
CARÁCTER	Anual	CURSO ACADÉMICO	1998/99
DURACIÓN APROXIMADA	2 horas		

APELLIDOS: NOMBRE:

GRUPO: NÚMERO DE EXPEDIENTE:

```
Private Sub UserControl_Resize()  
    Command1.Move 0, 0, ScaleWidth, ScaleHeight  
End Sub  
  
' Escribir valores de propiedades en el almacenamiento  
Private Sub UserControl_WriteProperties(PropBag As PropertyBag)  
    Call PropBag.WriteProperty("FileName", m_FileName, m_def_FileName)  
    Call PropBag.WriteProperty("Caption", Command1.Caption, "Command1")  
End Sub  
  
Private Sub UserControl_Click()  
    RaiseEvent Click  
End Sub  
  
Public Property Get Caption() As String  
    Caption = Command1.Caption  
End Property  
  
Public Property Let Caption(ByVal New_Caption As String)  
    Command1.Caption() = New_Caption  
    PropertyChanged "Caption"  
End Property
```